

ΥΠΟΛΟΓΙΣΤΕΣ Ι

ΕΝΤΟΛΗ IF

Τι χρειάζεται η εντολή IF

- Μέχρι τώρα είδαμε ότι οι εντολές ενός προγράμματος εκτελούνται από την αρχή έως το τέλος διαδοχικά η μία μετά την άλλη.
- Υπάρχουν προγράμματα όπου χρειάζεται σε κάποια περίπτωση να εκτελεστούν ορισμένες εντολές και σε κάποια άλλη όχι.

Η εντολή **IF** επιτρέπει την επιλεκτική εκτέλεση κάποιων εντολών ελέγχοντας μια συνθήκη.

1

2

Παράδειγμα #1

Κατασκευάστε πρόγραμμα που θα επιλύει την εξίσωση πρώτου βαθμού
 $Ax+B=0$
όταν δίνονται τα A και B.

Υπενθύμιση: Η λύση είναι $-\frac{B}{A}$ αν $A \neq 0$
αλλιώς δεν υπάρχει λύση.

Παράδειγμα #1

```
PROGRAM SOLVE1
  IMPLICIT NONE
  DOUBLE PRECISION A, B, X

  WRITE (*,*) 'Εισάγετε τους συντελεστές A,B'
  READ (*,*) A, B

  IF ( A /= 0 ) THEN
    X = -B/A
    WRITE (*,*) 'Η λύση είναι: ', X
  ELSE
    WRITE (*,*) 'Δεν υπάρχει λύση'
  END IF
```

END

3

4

Συντακτικό της εντολής IF (1/4)

```
IF ( σύγκριση ) THEN
    εντολή1
    εντολή2
    ⋮
ELSE
    εντολή1
    εντολή2
    ⋮
END IF
```

Σημείωση: κάθε εντολή γράφεται μερικά κενά (ή ένα tab) πιο δεξιά για ευκρίνεια.

5

Τι συγκρίσεις μπορεί να γίνουν ;

Μεταξύ δύο μεταβλητών ή γενικότερα δύο παραστάσεων **A** και **B** μπορούν να γίνουν οι παρακάτω συγκρίσεις:

Σύγκριση	Τι σημαίνει	Παλαιότερα	Ετυμολογία
$A == B$	$A = B$	<code>A .EQ. B</code>	Equal
$A /= B$	$A \neq B$	<code>A .NE. B</code>	Not Equal
$A > B$	$A > B$	<code>A .GT. B</code>	Greater Than
$A < B$	$A < B$	<code>A .LT. B</code>	Less Than
$A >= B$	$A \geq B$	<code>A .GE. B</code>	Greater or Equal
$A <= B$	$A \leq B$	<code>A .LE. B</code>	Less or Equal

Κάθε σύγκριση είναι είτε **αληθής** είτε **ψευδής**.

6

Πως λειτουργεί η εντολή IF

```
IF ( σύγκριση ) THEN
    εντολή1
    εντολή2
    ⋮
ELSE
    εντολή1
    εντολή2
    ⋮
END IF
```

Πρώτα γίνεται η **σύγκριση**

Εάν η **σύγκριση** είναι **αληθής** εκτελείται το πρώτο τμήμα εντολών και το πρόγραμμα συνεχίζει μετά το **END IF**.

Εάν η **σύγκριση** είναι **ψευδής** εκτελείται το δεύτερο τμήμα εντολών.

Σημείωση: Το τμήμα ELSE μπορεί να παραληφθεί.

7

Παράδειγμα #2

Κατασκευάστε πρόγραμμα που θα βρίσκει το μεγαλύτερο από δύο αριθμούς A, B που θα εισάγονται από το πληκτρολόγιο.

8

Παράδειγμα #2

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

```
PROGRAM LARGE
  IMPLICIT NONE
  DOUBLE PRECISION A, B

  WRITE (*,*) 'Εισάγετε δύο αριθμούς A, B'
  READ (*,*) A, B

  IF ( A > B ) THEN
    WRITE (*,*) 'Μεγαλύτερος είναι ο ', A
  ELSE
    WRITE (*,*) 'Μεγαλύτερος είναι ο ', B
  END IF
END
```

9

Παράδειγμα #2 (2^{ος} τρόπος)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

```
PROGRAM LARGE
  IMPLICIT NONE
  DOUBLE PRECISION A, B, M

  WRITE (*,*) 'Εισάγετε δύο αριθμούς A, B'
  READ (*,*) A, B

  IF ( A > B ) THEN
    M = A
  ELSE
    M = B
  END IF
  WRITE (*,*) 'Ο μεγαλύτερος είναι ο ', M
END
```

10

Παράδειγμα #2 - Επέκταση

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Πως θα βρούμε το μεγαλύτερο από **τρεις** αριθμούς A, B, C που θα εισάγονται από το πληκτρολόγιο ;




Σκεφτείτε το

11

Παράδειγμα #3

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Κατασκευάστε πρόγραμμα που θα βρίσκει ποιο από τα δύο σημεία του επιπέδου (x_1, y_1) και (x_2, y_2) βρίσκεται πιο μακριά και πόσο από το σημείο (x_0, y_0) .



Υπενθύμιση:

Η απόσταση d_1 είναι:

$$d_1 = \sqrt{(x_1 - x_0)^2 + (y_1 - y_0)^2}$$

12

Παράδειγμα #3

```
PROGRAM DIST
  IMPLICIT NONE
  DOUBLE PRECISION X0, Y0, X1, Y1, X2, Y2
  DOUBLE PRECISION D1, D2


  WRITE (*,*) 'Εισάγετε τα σημεία'
  READ (*,*) X0, Y0, X1, Y1, X2, Y2

  D1 = SQRT((X1-X0)**2+(Y1-Y0)**2)
  D2 = SQRT((X2-X0)**2+(Y2-Y0)**2)
  IF ( D1 > D2 ) THEN
    WRITE (*,*) 'Μακρύτερα είναι το πρώτο', D1
  ELSE
    WRITE (*,*) 'Μακρύτερα είναι το δεύτερο', D2
  END IF
END
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Παράδειγμα #4

Κατασκευάστε πρόγραμμα που θα βρίσκει εάν ένα σημείο του επιπέδου (x_1, y_1) βρίσκεται εντός ή εκτός κύκλου με κέντρο (x_0, y_0) και ακτίνα R . Στην πρώτη περίπτωση το πρόγραμμα να εμφανίζει την **απόσταση από το κέντρο του κύκλου** ενώ στη δεύτερη την **απόσταση από την περιφέρειά του**.



ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

14

Παράδειγμα #4

```
PROGRAM CIRCLE
  IMPLICIT NONE
  DOUBLE PRECISION X0, Y0, R, X1, Y1, D

  WRITE (*,*) 'Εισάγετε κέντρο και ακτίνα κύκλου'
  READ (*,*) X0, Y0, R
  WRITE (*,*) 'Εισάγετε το σημείο'
  READ (*,*) X1, Y1

  D = SQRT((X0-X1)**2+(Y0-Y1)**2)
  IF ( D <= R ) THEN
    WRITE (*,*) 'Το σημείο είναι εντός ', D
  ELSE
    D = D-R
    WRITE (*,*) 'Το σημείο είναι εκτός ', D
  END IF
END
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

15

Συντακτικό της εντολής IF (2/4)

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

```
IF ( σύγκριση1 ) THEN
  εντολή
  ⋮
ELSE IF ( σύγκριση2 ) THEN
  εντολή
  ⋮
ELSE IF ( σύγκριση3 ) THEN
  εντολή
  ⋮
ELSE
  εντολή
  ⋮
END IF
```

} Τμήμα IF

} 1^ο τμήμα ELSE IF

} 2^ο τμήμα ELSE IF

} Τμήμα ELSE. Εκτελείται όταν όλες οι συγκρίσεις είναι ψευδείς.

16

Συντακτικό της εντολής IF (3/4)

Ορισμένες παρατηρήσεις για την εντολή IF:

- Το πρώτο τμήμα **IF** πρέπει να υπάρχει υποχρεωτικά.
- Το τμήμα **ELSE** μπορεί να παραληφθεί.
- Μπορεί να υπάρχουν πολλά τμήματα **ELSE IF**.
- Εκτελούνται οι εντολές ενός μόνο από όλα τα τμήματα μιας εντολής IF.
- Εντός κάθε τμήματος επιτρέπεται να υπάρχουν άλλες εντολές IF.

17

Συντακτικό της εντολής IF (4/4)

Η εντολή IF που περιγράψαμε λέγεται **τμηματική εντολή IF**. Σε ορισμένες περιπτώσεις δεν υπάρχουν τα τμήματα ELSE IF και ELSE, και επιπλέον το τμήμα IF περιέχει μια μόνο εντολή, δηλαδή είναι της μορφής:

```
IF ( σύγκριση ) THEN
    εντολή
END IF
```

Τότε η εντολή μπορεί να απλοποιηθεί σε μια γραμμή ως εξής:

```
IF ( σύγκριση ) εντολή
```

Η εντολή αυτή ονομάζεται **λογική εντολή IF**.

18

Παράδειγμα #5

Κατασκευάστε πρόγραμμα που θα βρίσκει τις πραγματικές λύσεις (αν υπάρχουν) της δευτεροβάθμιας εξίσωσης

$$Ax^2 + Bx + C = 0$$

όταν δίνονται τα A, B και C.

Υπενθύμιση:

- Αν A=0 τότε έχουμε μια εξίσωση πρώτου βαθμού.
- Οι λύσεις της δευτεροβάθμιας είναι:

$$x_{1,2} = \frac{-B \pm \sqrt{B^2 - 4AC}}{2A}$$

19

Παράδειγμα #5

```
PROGRAM SOLVE2
  IMPLICIT NONE
  DOUBLE PRECISION A, B, C, X, D, X1, X2

  WRITE (*,*) 'Εισάγετε τους συντελεστές A,B,C'
  READ (*,*) A, B, C

  IF ( A == 0 ) THEN
!   Είναι εξίσωση πρώτου βαθμού.
    IF ( B /= 0 ) THEN
      X = -C/B
      WRITE (*,*) ' Η λύση είναι: ', X
    ELSE
      WRITE (*,*) ' Δεν υπάρχει λύση'
    END IF
  END IF
```

Συνεχίζεται...

20

Παράδειγμα #5

(συνέχεια)

ELSE

! Είναι εξίσωση δευτέρου βαθμού.

D = B**2-4*A*C

IF (D == 0) THEN

X = -B/(2*A)

WRITE (*,*) ' Η λύση είναι: ', X

ELSE IF (D > 0) THEN

X1 = (-B+SQRT(D))/(2*A)

X2 = (-B-SQRT(D))/(2*A)

WRITE (*,*) ' Η λύσεις είναι:',X1,X2

ELSE

WRITE (*,*) ' Δεν υπάρχει λύση '

END IF

END IF

END

...Συνέχεια

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

21

Σύνθετες λογικές παραστάσεις

Μπορούμε να κατασκευάσουμε σύνθετες λογικές παραστάσεις με τη χρήση των **ΛΟΓΙΚΩΝ ΤΕΛΕΣΤΩΝ**
.OR. .AND. .NOT.

IF (σύγκριση1 .AND. σύγκριση2) THEN

IF (σύγκριση1 .OR. σύγκριση2) THEN

IF (.NOT. σύγκριση) THEN

Το αποτέλεσμα μιας σύνθετης λογικής παράστασης είναι είτε αληθές είτε ψευδές και εξαρτάται από τις επιμέρους συγκρίσεις.

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

22

Πίνακες αλήθειας (.AND.)

σύγκριση1	σύγκριση2	σύγκριση1 .AND. σύγκριση2
A	A	A
A	Ψ	Ψ
Ψ	A	Ψ
Ψ	Ψ	Ψ

Πρακτικός κανόνας:

Το τελικό αποτέλεσμα είναι αληθές όταν
και οι δύο
συγκρίσεις είναι αληθείς

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

23

Πίνακες αλήθειας (.OR.)

σύγκριση1	σύγκριση2	σύγκριση1 .OR. σύγκριση2
A	A	A
A	Ψ	A
Ψ	A	A
Ψ	Ψ	Ψ

Πρακτικός κανόνας:

Το τελικό αποτέλεσμα είναι αληθές όταν
είτε η μία είτε η άλλη
σύγκριση είναι αληθείς

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

24


Πίνακες αλήθειας (.NOT.)

σύγκριση	.NOT. σύγκριση
A	Ψ
Ψ	A

Ο τελεστής .NOT. **αντιστρέφει** το αποτέλεσμα της σύγκρισης.

Παράδειγμα #6

Με ποια εντολή IF μπορούμε να βρούμε αν μια μεταβλητή X είναι μεταξύ δύο δοσμένων ορίων A, B ;




```
IF (X >= A .AND. X <= B) THEN
```

Αληθής
Αληθής

➔ Αληθής


Παράδειγμα #6



```
IF (X >= A .AND. X <= B) THEN
```

Ψευδής
Αληθής

➔ Ψευδής



```
IF (X >= A .AND. X <= B) THEN
```

Αληθής
Ψευδής

➔ Ψευδής

Παράδειγμα #6 - Επέκταση

Με ποια εντολή IF μπορούμε να βρούμε αν μια μεταβλητή X είναι **εκτός** δύο δοσμένων ορίων A, B ;



Σκεφτείτε το

Προτεραιότητες λογικών τελεστών

Σε μια σύνθετη λογική παράσταση όπως π.χ.

```
A > 0 .AND. B > 0 .OR. C == 4
```

η σειρά των λογικών πράξεων καθορίζεται από την προτεραιότητα των τελεστών. Για τελεστές ίδιας προτεραιότητας οι πράξεις γίνονται από αριστερά προς τα δεξιά.

Τελεστής Προτεραιότητα

.NOT.	Υψηλή
.AND.	
.OR.	Χαμηλή

Οι πράξεις σε παρενθέσεις γίνονται πάντα πρώτες. Π.χ.

```
A > 0 .AND. (B > 0 .OR. C == 4)
```

29

Παράδειγμα #7

Με ποια εντολή IF μπορούμε να διαπιστώσουμε αν ένας ακέραιος είναι άρτιος ή περιττός ;

Υπενθύμιση:

Ένας ακέραιος είναι άρτιος αν είναι πολλαπλάσιο του δύο, δηλαδή διαιρείται ακριβώς με το δύο χωρίς να αφήνει υπόλοιπο.

30

Παράδειγμα #7

Έστω N ο ακέραιος.

```
K = N/2           ! Ακέραια διαίρεση με το 2.
Y = N-2*K         ! Υπόλοιπο της διαίρεσης.
IF ( Y == 0 ) THEN ! Έλεγχος του υπολοίπου.
    ... άρτιος ...
ELSE
    ... περιττός ...
END IF
```

ή πιο σύντομα:

```
IF (N-2*(N/2) == 0) THEN
```

31

Παράδειγμα #7 (συνάρτηση MOD)

Η συνάρτηση MOD(A,B) επιστρέφει ως αποτέλεσμα το υπόλοιπο της διαίρεσης του A από τον B.

```
IF ( MOD(N,2) == 0 ) THEN
    ... άρτιος ...
ELSE
    ... περιττός ...
END IF
```


Με την ίδια συνάρτηση μπορούμε να βρούμε αν ένας ακέραιος είναι π.χ. πολλαπλάσιο του 7:

```
IF ( MOD(N,7) == 0 ) THEN
```

32


Παράδειγμα #8

Δίνεται τετράγωνο με κέντρο το $(0,0)$ και πλευρά A . Κατασκευάστε πρόγραμμα που θα βρίσκει αν ένα σημείο (x,y) είναι εντός ή εκτός του τετραγώνου. Στην πρώτη περίπτωση θα εμφανίζει την απόσταση από το κέντρο των αξόνων ενώ στη δεύτερη την απόσταση από την πάνω δεξιά γωνία.



33

Παράδειγμα #8



Για να είναι ένα σημείο (x,y) εντός του τετραγώνου πρέπει:

Η συντεταγμένη x να είναι μεταξύ $-A/2$ και $A/2$

Επίσης η συντεταγμένη y να είναι μεταξύ $-A/2$ και $A/2$

34

Παράδειγμα #8

Πως βρίσκω αν το X είναι μεταξύ $-A/2$ και $A/2$;

```
IF ( X >= -A/2 .AND. X <= A/2 ) THEN
```

ή

```
IF ( ABS(X) <= A/2 ) THEN
```

Πως βρίσκω αν το Y είναι μεταξύ $-A/2$ και $A/2$;

```
IF ( Y >= -A/2 .AND. Y <= A/2 ) THEN
```

ή

```
IF ( ABS(Y) <= A/2 ) THEN
```

35

Παράδειγμα #8

Οι δύο έλεγχοι μπορούν να συνδυαστούν σε μία εντολή:

```
IF ( X >= -A/2 .AND. X <= A/2 .AND.  
Y >= -A/2 .AND. Y <= A/2 ) THEN
```

ή

```
IF ( ABS(X) <= A/2 .AND. ABS(Y) <= A/2 ) THEN
```

36

Παράδειγμα #8

```
PROGRAM SQUARE
  IMPLICIT NONE
  DOUBLE PRECISION X, Y, A, D
  WRITE (*,*) 'Εισάγετε την πλευρά του τετραγώνου'
  READ (*,*) A
  WRITE (*,*) 'Εισάγετε το σημείο'
  READ (*,*) X, Y
  IF ( X >= -A/2 .AND. X <= A/2 .AND.      &
      Y >= -A/2 .AND. Y <= A/2 ) THEN
    D = SQRT(X**2+Y**2)
    WRITE (*,*) 'Το σημείο είναι εντός ', D
  ELSE
    D = SQRT((A/2-X)**2+(A/2-Y)**2)
    WRITE (*,*) 'Το σημείο είναι εκτός ', D
  END IF
END
```

Παράδειγμα #8 – Επέκταση#1

Πως θα βρούμε αν ένα σημείο είναι εντός ή εκτός **παραλληλογράμμου** με κέντρο την αρχή των αξόνων και πλευρές A, B ;



Σκεφτείτε το

Παράδειγμα #8 – Επέκταση#2

Πως θα βρούμε αν ένα σημείο είναι εντός ή εκτός **παραλληλογράμμου** με κέντρο το σημείο (x_0, y_0) και πλευρές A, B ;



Σκεφτείτε το

Παράδειγμα #9


Δίνεται κύκλος ακτίνας R με κέντρο την αρχή των αξόνων και δύο σημεία με συντεταγμένες (x_1, y_1) και (x_2, y_2) .

Αν και τα δύο σημεία είναι εντός του κύκλου ή και τα δύο είναι εκτός του κύκλου βρείτε τη μεταξύ τους απόσταση.

Αν ένα σημείο είναι εντός και το άλλο είναι εκτός βρείτε ποιο από τα δύο απέχει λιγότερο από την περιφέρεια του κύκλου και ποια είναι η απόσταση αυτή.

Παράδειγμα #9

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF



Η διακεκομμένη γραμμή δείχνει ποια απόσταση ζητείται

41

Παράδειγμα #9

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Με ποια εντολή IF διαπιστώνω σε ποια από τις τρεις περιπτώσεις βρίσκομαι ;

```
A1 = SQRT(X1**2+Y1**2)
A2 = SQRT(X2**2+Y2**2)
IF (A1 <= R .AND. A2 <= R) THEN
    ...και τα δύο σημεία εντός...
ELSE IF (A1 > R .AND. A2 > R) THEN
    ...και τα δύο σημεία εκτός...
ELSE
    ...ένα σημείο εντός και ένα εκτός...
END IF
```

42

Παράδειγμα #9

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Όμως και στις δύο πρώτες περιπτώσεις ζητείται η απόσταση μεταξύ των δύο σημείων. Μπορώ να συνδυάσω τους δύο ελέγχους;


```
A1 = SQRT(X1**2+Y1**2)
A2 = SQRT(X2**2+Y2**2)
IF (A1 <= R .AND. A2 <= R .OR.
    A1 > R .AND. A2 > R) THEN
    ...και τα δύο σημεία εντός ή εκτός...
ELSE
    ...ένα σημείο εντός και ένα εκτός...
END IF
```

43

Παράδειγμα #9

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

Πως υπολογίζω τις αποστάσεις d_1 και d_2 ;



44

Παράδειγμα #9

Μπορώ να συνδυάσω τις δύο περιπτώσεις υπολογίζοντας τις αποστάσεις d_1 και d_2 ως:

```
D1 = ABS (R-A1)
D2 = ABS (A2-R)
```

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

45

Παράδειγμα #9

```
PROGRAM CIRC2
  IMPLICIT NONE
  DOUBLE PRECISION R, X1, Y1, X2, Y2
  DOUBLE PRECISION A1, A2, D, D1, D2

  WRITE (*,*) 'Εισάγετε τα R,X1,Y1,X2,Y2'
  READ (*,*) R, X1, Y1, X2, Y2
  A1 = SQRT (X1**2+Y1**2)
  A2 = SQRT (X2**2+Y2**2)

  IF (A1 <= R .AND. A2 <= R .OR.      &
      A1 > R .AND. A2 > R) THEN
    D = SQRT ((X1-X2)**2+(Y1-Y2)**2)
    WRITE (*,*) D
```

Συνεχίζεται...

ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

46

Παράδειγμα #9 (συνέχεια)

```
ELSE
  D1 = ABS (R-A1)
  D2 = ABS (A2-R)
  IF (D1 < D2) THEN
    D = D1
  ELSE
    D = D2
  END IF
  WRITE (*,*) D
END IF
```

...Συνέχεια

END


ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

47

Παράδειγμα #10

Δίνεται τετράγωνο πλευράς 2 καθώς και ο εγγεγραμμένος και ο περιγεγραμμένος κύκλος. Δεδομένου ενός σημείου (x,y) βρείτε αν το σημείο αυτό:

- Είναι εντός του εγγεγραμμένου κύκλου (πορτοκαλί περιοχή).
- Είναι εντός του τετραγώνου αλλά εκτός του εγγεγραμμένου κύκλου (λευκή περιοχή).
- Είναι εντός του περιγεγραμμένου κύκλου αλλά εκτός του τετραγώνου (γαλάζια περιοχή).
- Είναι εκτός του περιγεγραμμένου κύκλου.





ΥΠΟΛΟΓΙΣΤΕΣ I - ΕΝΤΟΛΗ IF

48

Παράδειγμα #10

Ορισμένες παρατηρήσεις:

- Οι συντεταγμένες των τεσσάρων γωνιών του τετραγώνου είναι: 
- Η ακτίνα του εγγεγραμμένου κύκλου είναι 1
- Η ακτίνα του περιγεγραμμένου κύκλου είναι $\sqrt{2}$



49

Παράδειγμα #10

```

PROGRAM INAREA
  IMPLICIT NONE
  DOUBLE PRECISION X, Y

  WRITE (*,*) 'Εισάγετε τα X, Y'
  READ (*,*) X, Y

  IF (SQRT(X**2+Y**2) <= 1) THEN
    WRITE (*,*) 'Εντός του εγγεγραμμένου κύκλου'
  ELSE IF (X >= -1 .AND. X <= 1 .AND.      &
           Y >= -1 .AND. Y <= 1) THEN
    WRITE (*,*) 'Εντός του τετραγώνου'
  ELSE IF (SQRT(X**2+Y**2) <= SQRT(2.D0)) THEN
    WRITE (*,*) 'Εντός του περιγεγραμμένου κύκλου'
  ELSE
    WRITE (*,*) 'Εκτός του περιγεγραμμένου κύκλου'
  END IF

END
  
```